

---

---

## Algorithms

---

---

### MENU: A Menu Creating Word

*Contributed by Kevin Appert, Lockheed Missiles & Space Co., Inc.,  
Space Sciences Laboratory, 3251 Hanover Street, Palo Alto, CA. 94304*

#### *Abstract*

A technique for generating a menu from a list of FORTH words is presented.

The usual interface that a user encounters in a FORTH application is an OK prompt. The user is expected to type a FORTH word to get what he wants from the application. This is very often not possible for someone who is not a professional FORTH programmer. One simple solution is to provide a list of the FORTH words in the application from which the user can select the one he wants (still by typing it in). As a step beyond this, the application's vocabulary can be sealed so that the user is protected from using any of the rest of the FORTH words by mistake.

The word **MENU:** is a slightly more elegant approach. At run time it prints a list of FORTH words, hopefully with very descriptive names. The user moves the cursor to his choice and then presses the return key to execute the chosen word.

The code in the accompanying screens was implemented on a PDP-11 running FIG-FORTH. The terminal used was a VT-100 with ANSI standard escape sequences for cursor positioning.

The word **PAGE** clears the screen. The definition of **ARROW** would have to be changed to accommodate other terminals. Also, it would be easy to change the definition of **MENU:** to accommodate a terminal that does not provide absolute cursor positioning.

One possible extension would involve the use of an alternate key, such as Escape or Help, to select not the execution of the word, but a short description of what it will do when selected.

#### *References*

- Laxen, Henry "Screen-Oriented Editor in FORTH", *Dr. Dobb's Journal*, Sept. 1981, p. 27.
- Eaker, Charles Dr. "Case Control Structure", *Forth Dimensions*, Vol. II, No. 3, p. 37.
- Laxen, Henry "Defining Words II" *Forth Dimensions*, Vol. IV, No. 2, p. 20.

Manuscript received September 1984.

```

SCR # 9 KLA 07/08/83
0 \ case
1 : case ?comp csp @ ! csp 4 ; immediate
2
3 : of      4 ?pairs compile over compile = compile 0branch
4     here 0 , compile drop 5 ; immediate
5
6 : endof  5 ?pairs compile branch here 0 ,
7     swap 2 [compile] endif 4 ; immediate
8
9 : endcase 4 ?pairs compile drop begin sp@ csp @ = 0=
10     while 2 [compile] endif repeat csp ! ; immediate
11
12 \ THIS IS DR. EAKER'S CASE STATEMENT FROM F.D.II/3p37
13
14
15

```

```

SCR # 10 KLA 07/08/83
0 \ CRTXY KEY
1
2 : CRTXY ( X-POSITION Y-POSITION --- )
3     SEDITOR CRTXY ;          FORTH
4 \ CRTXY IS AN ALIAS FOR THE CURSOR POSITIONING WORD IN THE
5 \ SCREEN EDITOR VOCABULARY
6 \ SAME AS IN HENRY LAXEN'S SCREEN EDITOR IN D.D.J. #59
7
8 : KEY SKEY ;
9 \ IN PDP-11 FIG FORTH, THE WORD KEY READS IN A WHOLE LINE
10 \ AND NOT A SINGLE KEY, SO IT HAS TO BE REPLACED WITH SKEY
11 \ WHICH GETS A SINGLE KEYSTROKE
12
13 : 2+ 2 + ;
14
15

```

```

SCR # 11 KLA 06/29/83
0 \ VARIABLES
1
2 0 VARIABLE MENU.COUNT
3 \ MENU.COUNT WILL HOLD THE NUMBER OF ENTRIES IN THE MENU
4 \ CURRENTLY DISPLAYED
5
6 0 VARIABLE MENU.POINTER
7 \ KEEPS TRACK OF THE LINE THAT THE CURSOR IS ON
8
9 0 VARIABLE QUIT.FLAG
10 \ SET INSIDE THE CASE STATEMENT IN SELECT TO INDICATE THAT
11 \ A CHOICE HAS BEEN MADE
12
13
14
15

```

```

SCR # 12
0 \ ?POINT ARROW
1 : ?POINT ( --- )
2     MENU.POINTER @ DUP 1 < SWAP MENU.COUNT @ > OR
3     IF ( OUT OF BOUNDS ) 1 MENU.POINTER ! 5 1 CRTXY THEN ;
4 \ ?POINT CHECKS TO SEE IF THE USER HAS WALKED THE CURSOR PAST
5 \ THE BEGINNING OR END OF THE MENU AND IF SO PUTS HIM BACK AT
6 \ THE FIRST CHOICE
7
8 : ARROW ( --- )      KEY KEY SWAP
9     91 = IF ( COULD BE AN ARROW )
10    CASE 66 OF 27 EMIT 91 EMIT 66 EMIT ( DOWN)
11        1 MENU.POINTER +! ?POINT ENDOF
12    65 OF 27 EMIT 91 EMIT 65 EMIT ( UP)
13        -1 MENU.POINTER +! ?POINT ENDOF
14    BEEP ( NOT UP OR DOWN ) ENDCASE THEN ;
15 \ ARROW CHANGES CURSOR POSITION AND MENU.POINTER

```

KLA 07/08/83

```

SCR # 13
0 \ MENU-LIST
1 : MENU-LIST ( FIRST-ADDRESS --- )
2     PAGE ." USE UP AND DOWN ARROWS TO MOVE CURSOR AND RETURN
3 TO SELECT" CR 0 MENU.COUNT !
4     BEGIN DUP @ DUP
5     [ ' ;S CFA ] LITERAL = IF 1 ( FOUND ;S , SET END)
6     ELSE 5 SPACES 2+ NFA ID. CR
7     1 MENU.COUNT +!
8     2+ ( POINT AT NEXT ENTRY IN MENU )
9     0 ( HAVE NOT FOUND ;S YET SO GO ON)
10    THEN
11    UNTIL DROP DROP      5 1 CRTXY
12    1 MENU.POINTER ! ( PUT CURSOR AND POINTER AT START ) ;
13
14 \ LIST THE MENU , COUNT THE ENTRIES , THEN HOME THE CURSOR
15

```

KLA 06/30/83

```

SCR # 14
0 \ SELECT
1 : SELECT ( FIRST-ADDRESS --- ) 0 QUIT.FLAG !
2     BEGIN KEY
3     CASE 13 OF 1 QUIT.FLAG ! ENDOF ( MADE CHOICE)
4     27 OF ARROW ENDOF
5     BEEP ( NOT A RETURN OR ESCAPE CHARACTER)
6     ENDCASE
7     QUIT.FLAG @ UNTIL
8     PAGE MENU.POINTER @ 1 -2 * + @ EXECUTE ;
9
10 \ SELECT IS A LOOP THAT MOVES THE CURSOR UP AND DOWN UNTIL A
11 \ CHOICE IS MADE AND THEN EXECUTES THE CHOICE
12
13
14
15

```

SCR # 15

KLA 07/30/83

```

0 \ MENU:
1 : MENU: ( ---)
2     <BUILDS ] SMUDGE DOES> DUP MENU-LIST SELECT ;
3
4 \ MENU: MAKES A MENU FROM A LIST OF FORTH WORDS. AT COMPILE
5 \ TIME IT MAKES A LIST OF EXECUTION VECTORS. JUST LIKE A
6 \ FORTH COLON DEFINITION, OR HENRY LAXEN'S CASE STATEMENT
7 \ FROM D.D.J.#59P32 FROM WHICH IT WAS BORROWED. AT RUN-TIME
8 \ IT PRESENTS A LIST OF THE FORTH WORDS AND THE USER MOVES
9 \ THE CURSOR WITH THE UP AND DOWN KEYS AND SELECTS WITH THE
10 \ RETURN KEY.
11
12
13
14
15

```

SCR # 16

KLA 06/28/83

```

0 \ MENU-SAMPLE
1 \ HERE IS A SAMPLE OF THE USE OF MENU:
2 MENU: MENU-SAMPLE ( --- )
3     PAGE VLIST BYE ;
4
5
6
7
8
9
10
11
12
13
14
15

```

SCR # 17

KLA 07/30/83

```

0 \ LOAD SCREEN
1 : MENUSTART ;
2 9 LOAD
3 10 LOAD
4 11 LOAD
5 12 LOAD
6 13 LOAD
7 14 LOAD
8 15 LOAD
9 16 LOAD
10
11
12
13
13
14
15

```